

**DEVIATION SURVEY CALCULATIONS
USING CURVATURE METHOD**

DSCM PROGRAM



Author

Dr. Ziad Amer Hammoodi

Ex Technical Team Leader –ADMA-OPCO

Ex. Reservoir Engineering Software Specialist - ADNOC

1- INTRODUCTION

The **Curvature** method is probably the most accurate method for deviation survey calculations. Unfortunately, this method is complicated and difficult to understand. A scientific computer is required to implement it.

The main source of **calculation errors** in all methods of calculations is due to the fact that the calculation at one survey point depends on the results from the previous survey point. Therefore, calculation errors will be accumulated until the last survey point.

Historically, most of **Offshore Operating Companies** adopted the curvature method due to the fact that their drilled wells were mostly deviated as many wells could share the same Platform.

In the following sections, we first start by the description of the **well path** and the transformation of survey point coordinates from the measured **geological coordinates** system into the calculated local **geographical coordinates** system.

The **curvature method** will be also explained in a comprehensive and original approach never seen in the literature. In this method the **rate of curvature** of the curved well path and the variations in measured deviation and azimuth angles between each two successive survey points will be considered.

A simplified **FORTRAN** program is developed based on this method. The program can be used as an accurate computing tool of calculations for new deviated and horizontal wells and to check calculated results of third party software packages or provided by service companies

2- WELL PATH

The **well path** is determined by survey points measured on the trajectory of a well during drilling operations.

A survey point on the well path trajectory can be presented in the space by a **geological coordinates system (I,A,M)** or by a **geographical coordinates system (E,N,V)**.

The first coordinates system is similar to spherical coordinates system, in which a point in the space is represented by a distance and two angles. The **geological measured coordinates (i,a,m)** are:

<i>i</i>	measured inclination angle (in degrees) from vertical
<i>a</i>	<i>measured</i> azimuth angle (in degrees) on the horizontal plane
<i>m</i>	measured depth (in feet)

The origin **(0,0,0)** of the measured **geological coordinates system** is the reference point called the **tie point** defined by the penetration of well path at Rotary Table level. This is the starting point for measuring depth.

The second coordinates system is a local rectangular coordinates system and the **geographical calculated coordinates (e,n,v)** are:

<i>e</i>	easting displacement from the vertical in the horizontal plane (in feet)
<i>n</i>	northing displacement from the vertical in the horizontal plane (in feet)
<i>v</i>	True V ertical Depth, TVD (in feet below Rotating Table)

The origin **(0,0,0)** of the calculated **geographical coordinates system** is the same reference point of the **geological coordinates system**.

The well deviation survey calculations convert the measured **geological coordinates** **(i,a,m)** for each survey point encountered along the well path trajectory into local **geographical coordinates** **(e,n,v)**.

Figure (1) shows the graphical representation of coordinate systems. The rotary table level (RTE) and the Mean Sea Level (MSL) are also shown. A third coordinate system called regional **geographical coordinates system** **(X,Y,Z)** is also used to represent the calculated UTM (Universal Transverse Mercator) and the Sub-Sea TVD coordinates **(x,y,z)** where:

- x** UTM coordinate in the easting direction (in meters)
- y** UTM coordinate in the northing direction (in meters)
- z** True Vertical Depth, TVD (in feet Sub-Sea)

This third coordinate system will not be used in this report. It has been covered in a more detailed report covering tangential and curvature methods (see reference 8).




Fig. 1 - Graphical Representation of Coordinate Systems

Deviation survey calculations are performed by considering **two successive survey points (1 & 2)**. Knowing the coordinates of the first point, the coordinates of the second point are calculated. This procedure is repeated until the latest survey point.

The **inaccuracies** and **discrepancies** encountered among different calculations methods is mainly due to the fact that any calculation error or inaccuracy on one survey point will be perpetuated on all successive survey points.

The **geographical coordinates** **(e₂, n₂, v₂)** for the survey point **(2)** along the well path trajectory are calculated the from the measured **geological coordinates** **(i₂, a₂, m₂)** at the same survey point and both the measured coordinates **(i₁, a₁, m₁)** and the calculated coordinates **(e₁, n₁, v₁)** at the previous survey point **(1)**.

The **well path segment** (Figure 2) shows the increments of the geological and geographical coordinates between two successive survey points (**1 & 2**) as follows:

$\Delta i = i_2 - i_1$	increment in inclination angle
$\Delta a = a_2 - a_1$	increment in azimuth angle
$\Delta m = m_2 - m_1$	increment in measured depth
$\Delta e = e_2 - e_1$	increment in easting displacement
$\Delta n = n_2 - n_1$	increment in northing displacement
$\Delta v = v_2 - v_1$	increment in TVD

The incremental symbol (Δ) will be used in **trigonometry and vector theory**. The well path segment will be considered as a straight line and the variation of inclination and azimuth angles along the well path segment will be ignored.

The differential symbol (d) will be used in **differentiation and integration theory**. The well path segment be considered as a curve and the variation of inclination and azimuth angles along the well path segment will be considered.

$d i = i_2 - i_1$	differential in inclination angle
$d a = a_2 - a_1$	differential in azimuth angle
$d m = m_2 - m_1$	differential in measured depth
$d e = e_2 - e_1$	differential in easting displacement
$d n = n_2 - n_1$	differential in northing displacement
$d v = v_2 - v_1$	differential in TVD




Fig. 2 - Graphical Representation of a Well Path Segment

3- CURVATURE METHOD

This method is based on calculation of the **total rate of curvature (K)** of the well path and on the **variations of the inclination and azimuth angles** along the well path segment between two successive survey points 1 & 2.

By definition, the **total rate of curvature (K)** of the well path is expressed by the rate of change of curvature angle (in radian) per unit of measured depth.

By definition, the **Dog-Leg Severity (K_d)** is equal to the rate of change of curvature angle (in degrees) between both ends of a well path segment of 100 feet.

$$K_d = 100 K (180 / \pi)$$

We will demonstrate that total rate of curvature (**K**) and the Dog-Leg Severity (**K_d**) can be expressed in terms of two other factors **K_i** & **K_a** as follows:

$$K = \text{sqrt} (K_i^2 + K_a^2)$$

There are many variations of the curvature method, as radius of curvature method and minimum of curvature method. We will demonstrate an advanced curvature method used within one of the operating companies for historical consistency.

This curvature method is probably the most accurate but the most complex method due to the fact that the curvature and the variation of angles not only at both sides of the well path segment but along the path are taken in consideration.

3.1- Dog-Leg Severity in Curvature Method

The total rate of curvature of a curved well path segment between two survey points 1 and 2 (see Figure 2) in the **geographical coordinates system (E,N,V)** can be defined mathematically as follows:

$$K = \text{sqrt} [(d^2e/dm^2)^2 + (d^2n/dm^2)^2 + (d^2v/dm^2)^2] \quad (1)$$

The differentials of **e** and **n** in respect to horizontal displacement **d** are as follows:

$$de / dd = \sin a \quad (2)$$

$$dn / dd = \cos a \quad (3)$$

The differentials of **e** and **n** with respect to measured depth **m** are as follows:

$$de / dm = (de / dd)(dd / dm) = \sin a \sin i \quad (4)$$

$$dn / dm = (dn / dd)(dd / dm) = \cos a \sin i \quad (5)$$

Additionally, the differential of **v** with respect to measured depth **m** is as follows:

$$dv / dm = \cos i \quad (6)$$

By calculating the second derivatives of **e**, **n** and **v** with respect to **m**, we get:

$$d^2e/dm^2 = \sin i \cos a (da / dm) + \sin a \cos i (di / dm) \quad (7)$$

$$d^2n/dm^2 = - \sin i \sin a (da / dm) + \cos a \cos i (di / dm) \quad (8)$$

$$d^2v/dm^2 = - \sin i (di / dm) \quad (9)$$

Let us introduce the following two rate of angle changes:

$$K_i = di / dm \quad (10)$$

$$K_a = (da / dm) \sin i \quad (11)$$

By substitution of K_i and K_a into (7), (8), and (9) we obtain the following:

$$d^2e / dm^2 = \cos a K_a + \sin a \cos i K_i \quad (12)$$

$$d^2n / dm^2 = - \sin a K_a + \cos a \cos i K_i \quad (13)$$

$$d^2v / dm^2 = - \sin K_i \quad (14)$$

By substituting the resulted terms from (12), (13), and (14) into (1) and by simplifying we obtain the following simplified formula:

$$K = \text{sqrt} (K_i^2 + K_a^2) \quad (15)$$

where:

K total rate of curvature (in radian/feet)

K_i rate of change of inclination angle per unit of measured depth

K_a rate of change of azimuth angle per unit of measured depth multiplied by **sin i**

The Dog-Leg Severity in degrees per 100 feet of measured depth will be:

$$K_d = 100 K (180 / \pi) \quad (16)$$

3.2- Calculations Cases

Following the **variations of the inclination angle and azimuth angle** along a well path segment between two successive survey points (1 & 2) the two factors **K_i** & **K_a** and the change in geographical coordinates (**d e**, **d n**, **d v**) will be expressed differently. There are different four cases of calculations to be considered as follows:

Case 1: constant inclination ($i_1 = i_2$)

constant azimuth ($a_1 = a_2$)

Case 2: constant inclination ($i_1 = i_2$)

varying azimuth ($a_1 \neq a_2$)

Case 3: varying inclination ($i_1 \neq i_2$)

constant azimuth ($a_1 = a_2$)

Case 4: varying inclination ($i_1 \neq i_2$)

varying azimuth ($a_1 \neq a_2$)

For each case, the calculations of **K_i**, **K_a**, **d e**, **d n**, **d v** will be provided in the following sections. **K_i** & **K_a** will be calculated first. Then **d e**, **d n**, **d v** will be calculated making use of calculated **K_i** & **K_a** when necessary.

The following general formulas will be applied on all four cases to calculate the Dog-Leg severity at each survey point.

$$K = \text{sqrt} (K_i^2 + K_a^2)$$

$$K_d = 100 K (180 / \pi)$$

rate of Curvature angle in radian per foot

Dog-Leg severity in degrees per 100 feet

We will use equally differential symbol (**d**) and the increment symbol (**Δ**) for the measured depth and the geographical coordinates between two successive survey points in all cases.

However, we use only differential symbol (**d**) for inclination and azimuth angles for the purpose of integration between both ends of the well path segment when required.

The following general formulas will be applied on all four cases to calculate the geographical coordinates at survey point 2

$$\begin{aligned} e_2 &= e_1 + \Delta e && \text{easting coordinate of second survey point} \\ n_2 &= n_1 + \Delta n && \text{northing coordinate of second survey point} \\ v_2 &= v_1 + \Delta v && \text{vertical coordinate of second survey point} \end{aligned}$$

3.2.1- Case 1

$$\begin{aligned} \text{constant inclination} & \quad (i_1 = i_2) \\ \text{constant azimuth} & \quad (a_1 = a_2) \end{aligned}$$

The well path is a straight line. From Figure (1), and by using equally the differential symbol (**d**) and the incremental symbol (**Δ**) we can write the following:

$$\begin{aligned} K_i &= 0 \\ K_a &= 0 \end{aligned}$$

$$d e = d m \sin i_1 \sin a_1 \quad (17)$$

$$d n = d m \sin i_1 \cos a_1 \quad (18)$$

$$d v = d m \cos a_1 \quad (19)$$

$$\Delta e = (m_2 - m_1) \sin i_1 \sin a_1 \quad (20)$$

$$\Delta n = (m_2 - m_1) \sin i_1 \cos a_1 \quad (21)$$

$$\Delta v = (m_2 - m_1) \cos i_1 \quad (22)$$

We notice that this case is identical to tangential method regarding the geographical coordinates but not for Dog-Leg severity which will be null in this case.

3.2.2- Case 2

$$\begin{aligned} \text{constant inclination} & \quad (i_1 = i_2) \\ \text{varying azimuth} & \quad (a_1 \neq a_2) \end{aligned}$$

$$\begin{aligned} K_i &= 0 \\ K_a &= (a_2 - a_1) / (m_2 - m_1) \end{aligned}$$

As azimuth angle is varying along the well path segment we replace the following terms into (17) and (18) respectively:

$$d m \sin a_1 \text{ by the integrals } \int d m \sin a \text{ between } a_1, a_2$$

$$d m \cos a_1 \text{ by the integrals } \int d m \cos a \text{ between } a_1, a_2$$

Then, from (11), we replace **d m** by **sin i₁ d a / K_a** and perform the two integrations as follows:

$$\begin{aligned}
d e &= \sin i_1 \int d m \sin a \\
&= (\sin i_1 \sin i_1 / K_a) \int \sin a d a \\
d n &= \sin i_1 \int d m \cos a \\
&= (\sin i_1 \sin i_1 / K_a) \int \cos a d a \\
d v &= d m \cos a_1 \\
\Delta e &= \sin i_1 \sin i_1 (\cos a_1 - \cos a_2) / K_a & (23) \\
\Delta n &= \sin i_1 \sin i_1 (\cos a_2 - \cos a_1) / K_a & (24) \\
\Delta v &= (m_2 - m_1) \cos i_1 & (25)
\end{aligned}$$

3.2.3- Case 3

varying inclination $(i_1 \neq i_2)$
constant azimuth $(a_1 = a_2)$

$$\begin{aligned}
K_i &= (i_2 - i_1) / (m_2 - m_1) \\
K_a &= 0
\end{aligned}$$

As inclination is varying along the well path segment we replace the following term into (17) and (18) respectively:

$$d m \sin i_1 \text{ by the integral } \int d m \sin i \text{ between } i_1, i_2$$

Then, from (10), we replace $d m$ by $d i / K_i$ and perform the integration as follows:

$$\begin{aligned}
d e &= \sin a_1 \int d m \sin i \\
&= (\sin a_1 / K_i) \int \sin i d i \\
d n &= \sin a_1 \int d m \sin i \\
&= (\cos a_1 / K_i) \int \sin i d i \\
d v &= \int \cos i d i / K_i \\
\Delta e &= \sin a_1 (\cos i_1 - \cos i_2) / K_i & (26) \\
\Delta n &= \cos a_1 (\cos i_2 - \cos i_1) / K_i & (27) \\
\Delta v &= (\sin i_2 - \sin i_1) / K_i & (28)
\end{aligned}$$

3.2.4- Case 4

varying inclination $(i_1 \neq i_2)$
varying azimuth $(a_1 \neq a_2)$

$$\begin{aligned}
K_i &= (i_2 - i_1) / (m_2 - m_1) \\
K_a &= K_i (a_2 - a_1) / [\text{Log} (\tan (i_2 / 2)) - \text{Log} (\tan (i_1 / 2))]
\end{aligned}$$

K_a is derived by dividing (10) over (11), rearranging the terms and integrating both sides between i_1, i_2 and a_1, a_2 . K_a will be expressed as function of K_i and changes in a and i as follows:

$$\begin{aligned}
K_i/K_a &= (d i / d m) / ((d a / d m) \sin i) \\
K_i/K_a &= (d i / d a) \operatorname{cosec} i \\
\int d a &= (K_a / K_i) \int (\operatorname{cosec} i) d i \\
K_a &= K_i (a_2 - a_1) / [\operatorname{Log} (\tan (i_2 / 2)) - \operatorname{Log} (\tan (i_1 / 2))] \quad (29)
\end{aligned}$$

As the inclination and the azimuth are both varying along the well path we replace the following terms into (17) and into (18) respectively:

$$\begin{aligned}
d m \sin i_1 \sin a_1 &\text{ by the integral } \int d m \sin a \sin i \text{ between } i_1, i_2 \\
d m \sin i_1 \cos a_1 &\text{ by the integral } \int d m \cos a \sin i \text{ between } i_1, i_2
\end{aligned}$$

Then, from (10), we replace $d m$ by $d i / K_i$ and perform the two integrations by using **Simpson's Rule**.

Simpson's Rule permits to calculate the integral $\int f(x) dx$ in the interval $[a, b]$ for the function $y = f(x)$ defined at three points (x_0, y_0) , (x_1, y_1) , (x_2, y_2) , where $x_0 = a$, $x_1 = (a + b)/2$, $x_2 = b$ and $h = (b - a)/2$, by the following expression:

$$\int f(x) dx = (h/3) (y_0 + 4y_1 + y_2)$$

In our case, we consider the three survey points (i_1, a_1, y_1) , (i_h, a_h, y_h) and (i_2, a_2, y_2) where $y = f(i, a)$ is the function to be integrated. The point (i_h, a_h, y_h) is considered situated half way between the other measured survey points. i_h and a_h can be expressed as follows:

$$i_h = (i_1 + i_2) / 2 \quad (30)$$

$$a_h = a_1 + (K_a / K_i) [\operatorname{Log} (\tan (i_h / 2)) - \operatorname{Log} (\tan (i_1 / 2))] \quad (31)$$

Applying **Simpson's Rule** using $h = (i_2 - i_1) / 2$ we get the following :

$$\begin{aligned}
d e &= \int d m \sin a \sin i \\
&= (1 / K_i) \int \sin a \sin i d i \\
&= (h / 3 K_i) [\sin i_1 \sin a_1 + 4 \sin i_h \sin a_h + \sin i_2 \sin a_2]
\end{aligned}$$

$$\begin{aligned}
d n &= \int d m \cos a \sin i \\
&= (1 / K_i) \int \cos a \sin i d i \\
&= (h / 3 K_i) [\sin i_1 \cos a_1 + 4 \sin i_h \cos a_h + \sin i_2 \cos a_2]
\end{aligned}$$

$$d v = \int \cos i d i / K_i$$

$$\Delta e = [\sin i_1 \sin a_1 + 4 \sin i_h \sin a_h + \sin i_2 \sin a_2] (i_2 - i_1) / 6 K_i \quad (32)$$

$$\Delta n = [\sin i_1 \cos a_1 + 4 \sin i_h \cos a_h + \sin i_2 \cos a_2] (i_2 - i_1) / 6 K_i \quad (34)$$

$$\Delta v = (\sin i_2 - \sin i_1) / K_i \quad (35)$$

The calculation of Dog-Leg severity will be common in all four cases as follows:

$$K = \operatorname{sqrt} (K_i^2 + K_a^2) \quad (36)$$

$$K_d = 100 K (180 / \pi) \quad (37)$$

The calculate the geographical coordinates at survey point (2) will be also common in all for cases as will be see in the next section.

3.3- Formulation

The following table represents the summary of all formula which will be used in the Fortran program. For each case, the rate of change of inclination and azimuth (K_i, K_a) and the increments of coordinates between two survey points ($\Delta e, \Delta n, \Delta v$) are calculated. Then the Dog-Leg severity and the coordinates are calculated for any case.

CASE	FORMULAS
Case 1	$K_i = 0$ $K_a = 0$ $\Delta e = (m_2 - m_1) \sin i_1 \sin a_1$ $\Delta n = (m_2 - m_1) \sin i_1 \cos a_1$ $\Delta v = (m_2 - m_1) \cos i_1$
Case 2	$K_i = 0$ $K_a = (a_2 - a_1) / (m_2 - m_1)$ $\Delta e = \sin i_1 \sin i_1 (\cos a_1 - \cos a_2) / K_a$ $\Delta n = \sin i_1 \sin i_1 (\cos a_2 - \cos a_1) / K_a$ $\Delta v = (m_2 - m_1) \cos i_1$
Case 3	$K_i = (i_2 - i_1) / (m_2 - m_1)$ $K_a = 0$ $\Delta e = \sin a_1 (\cos i_1 - \cos i_2) / K_i$ $\Delta n = \cos a_1 (\cos i_2 - \cos i_1) / K_i$ $\Delta v = (\sin i_2 - \sin i_1) / K_i$
Case 4	$K_i = (i_2 - i_1) / (m_2 - m_1)$ $K_a = K_i (a_2 - a_1) / [\text{Log} (\tan (i_2 / 2)) - \text{Log} (\tan (i_1 / 2))]$ $i_h = (i_1 + i_2) / 2$ $a_h = a_1 + (K_a / K_i) [\text{Log} (\tan (i_h / 2)) - \text{Log} (\tan (i_1 / 2))]$ $\Delta e = [\sin i_1 \sin a_1 + 4 \sin i_h \sin a_h + \sin i_2 \sin a_2] (i_2 - i_1) / 6 K_i$ $\Delta n = [\sin i_1 \cos a_1 + 4 \sin i_h \cos a_h + \sin i_2 \cos a_2] (i_2 - i_1) / 6 K_i$ $\Delta v = (\sin i_2 - \sin i_1) / K_i$
All Cases	$K = \text{sqrt} (K_i^2 + K_a^2)$ $K_d = 100 K (180 / \pi)$ $e_2 = e_1 + \Delta e$ $n_2 = n_1 + \Delta n$ $v_2 = v_1 + \Delta v$

4- PROGRAMMING

4.1- DSCM Program Description

A Fortran program for deviation survey calculation **dscm** was developed using double precision. It consists of one main program **dscm.f** and one calculation subroutine **dsc.f** using the curvature method.

The main program **dscm.f** reads the input file, calls the subroutine **dsc.f** to perform the calculations, and finally print the output file containing the results. It prompts for the input file name which should be entered by the user.

This subroutine **dsc.f** uses the complicated formulation of the curvature method in a concise straightforward and simplified manner. This is due to the fact that the variables used in the formula and in the program are identical.

The program is compiled using Fortran 77 Compiler. It can be easily implemented and compiled on any computer with Fortran compiler as follows:

```
f77 dscm.f dscm.f -o @dscm
```

The **source code** of the main program and the calculation subroutine will be given in the next section. Concerned users can cut, paste and compile the program on their specific computer platforms.

4.2- Running DSCM Program

To run the **dscm** program , user can submit the following command from any workstation used as installation platform:

```
@dscm
```

Users can also copy test examples to the installation directory and test the program:

4.3- Input and Output files

The first line of the input file should contain the measured and the calculated coordinates of the tie-in point **(m,i,a, v,n,e)**. These items should be separated by one or more spaces.

The remaining lines of the input file should contain only the measured geological coordinates **(m,i,a)** for all survey points.

The name of input file is arbitrary. It is preferred to be the name of a well or a bore hole. The output file name will be identical to the input file name with the added extension **.out**.

The output file will list the measured and the calculated coordinates for all survey points as well as the Dog-Leg severity **(m,i,a,v,n,e,kd)**.

Input files and output files will be given as test examples of section 6.

5- PROGRAM SOURCE

```
PROGRAM DSCM
C*****
*****
C      I  :Array Of Inclination Angles.....(deg)
C      A  :Array Of Azimuth Angles.....(deg)
C      M  :Array Of Measured Depth.....(ft)
C      NP :Number Of Survey Points
C      E  :Array Of Easting Displacement.....(ft)
C      N  :Array Of Northing Displacement.....(ft)
C      V  :Array Of Vertical Depth Below RTKB.....(ft)
C      KD :Array Of Dog-Leg Severity.....(deg/100ft)
C*****
C***** DECLARATIONS:
      REAL*8 I(200),A(200),M(200)
      REAL*8 E(200),N(200),V(200),KD(200)
      CHARACTER*24 F10,F20
C***** OPENING OF INPUT/OUTPUT FILES *****
      WRITE(*,1)
1      FORMAT(1X,'ENTER survey points INPUT FILE NAME AND RETURN')
      READ(*,'(A)')F10
      OPEN(10,FILE=F10)
C*****CONCATENATION OF EXTENSION .out TO INPUT FILE NAME
      IC1=-1
      DO 2 IC=1,24
      IF(IC1.GT.0.AND.F10(IC:IC).EQ.' ') THEN
      IC2=IC-1
      GO TO 3
      ENDIF
      IF(IC1.EQ.-1.AND.F10(IC:IC).NE.' ') THEN
      IC1=IC
      ENDIF
2      CONTINUE
3      CONTINUE
      F20= F10(IC1:IC2)//'.'//'.out'
      OPEN(20,FILE=F20)
C***** READING STARTING TIE POINT *****
      NP=1
      READ(10,*,END=7)M(1),I(1),A(1),V(1),N(1),E(1),KD(1)
C***** READING SURVEY POINTS *****
4      NP=NP+1
      READ(10,*,END=5)M(NP),I(NP),A(NP)
      GO TO 4
5      IF(NP.EQ.1) GO TO 8
      NP=NP-1
C*****
      CALL DSC(M,I,A,V,N,E,KD,NP)
C*****
C WRITING OUTPUT RESULTS
      WRITE(20,*)' MD Incl. Azim. TVD NS EW ',
1 ' DLS'
      WRITE(20,6)(M(L),I(L),A(L),V(L),N(L),E(L),KD(L),L=1,NP)
6      FORMAT(1X,F6.0,2F7.2,F8.1,1X,2F8.1,1X,F5.2)
      GO TO 9
7      WRITE(*,*) 'WELL SURVEY HEADING IS NOT FOUND'
      GO TO 9
8      WRITE(*,*) 'WELL SURVEY STATIONS ARE NOT FOUND'
9      STOP
      END
```

```

SUBROUTINE DSC (M, I, A, V, N, E, KD, NP)
C*****
C*****
C INPUT:
C     I   :Array Of Inclination Angles.....(deg)
C     A   :Array Of Azimuth Angles.....(deg)
C     M   :Array Of Measured Depth.....(ft)
C     NP  :Number Of Survey Points
C OUTPUT:
C     E   :Array Of Easting Displacement.....(ft)
C     N   :Array Of Northing Displacement.....(ft)
C     V   :Array Of Vertical Depth Below RTKB (ft)
C     KD  :Array Of Dog_Leg Severity.....(deg/100ft)
C CONSTANTS:
C     DR  :Conversion Constant From Degree To Radian
C     R   :=0.001*DR The Smallest Value For Angles (rad)
C*****
C ****DECLARATIONS:
      REAL*8 I (200), A (200), M (200), I1, A1, M1, I2, M2, A2
      REAL*8 E (200), N (200), V (200), E1, N1, V1, E2, N2, V2
      REAL*8 KD (200), KI, KA, IH, AH, FM, DR
      DATA DR/0.0174533D0/
C*****
C*****INITIAL SURVEY AND REFERENCE POINT:
      M1=M(1)
      I1=I(1)*DR
      A1=A(1)*DR
      E1=E(1)
      N1=N(1)
      V1=V(1)
C*****
C*****STARTING CALCULATION LOOP:
      L=1
20    L=L+1
      IF(L.GT.NP) GO TO 9
      M2=M(L)
      I2=I(L)*DR
      A2=A(L)*DR
C*****
C*****CHECK THE DEPTH BETWEEN TWO SUCCESSIVE SURVEY POINTS:
      IF((M2-M1).GT.0.1D0) GO TO 21
      GO TO 9
C*****
C**** CHECK THE AZIMUTH BETWEEN TWO SUCCESSIVE SURVEY POINTS:
C**** REDUCE IF AZIMUTH DIFFERENCE IS OVER 180 DEGREE
21    IF(DABS(A2-A1).LE.180D0*DR) GO TO 23
      IF(A2.GE.A1) GO TO 22
      A2=A2+360.0D0*DR
      GO TO 21
22    A2=A2-360.0D0*DR
      GO TO 21
C*****
C**** CHECK THE INCLINATION FOR TWO SUCCESSIVE SURVEY POINTS:
C**** IF INCLINATION IS ZERO SET MEANINGLESS AZIMUTH TO CONSTANT
23    R=0.001D0*DR
      IF(I1.LT.R) A1=A2
      IF(I2.LT.R) A2=A1
C*****
C**** TEST FOR CASES OF CONSTANT AND VARYING INCLINATION AND AZIMUTH

```

```

IF((DABS(I2-I1).LT.R).AND.(DABS(A2-A1).LT.R)) GO TO 1
IF((DABS(I2-I1).LT.R).AND.(DABS(A2-A1).GE.R)) GO TO 2
IF((DABS(I2-I1).GE.R).AND.(DABS(A2-A1).LT.R)) GO TO 3
IF((DABS(I2-I1).GE.R).AND.(DABS(A2-A1).GE.R)) GO TO 4
C*****
C***** CASE 1: CONSTANT INCLINATION CONSTANT AZIMUTH
1   KI=0.D0
    KA=0.D0
    E2= E1 + (M2-M1)*DSIN(I1)*DSIN(A1)
    N2= N1 + (M2-M1)*DSIN(I1)*DCOS(A1)
    V2= V1 + (M2-M1)*DCOS(I1)
    GO TO 5
C*****
C***** CASE 2: CONSTANT INCLINATION VARYING AZIMUTH
2   KI=0.D0
    KA=(A2-A1)*DSIN(I1)/(M2-M1)
    E2= E1 + DSIN(I1)*DSIN(I1)*(DCOS(A1)-DCOS(A2))/KA
    N2= N1 + DSIN(I1)*DSIN(I1)*(DSIN(A2)-DSIN(A1))/KA
    V2= V1 + (M2-M1)*DCOS(I1)
    GO TO 5
C*****
C***** CASE 3 :VARYING INCLINATION CONSTANT AZIMUTH
3   KI=(I2-I1)/(M2-M1)
    KA=0.D0
    E2= E1 + DSIN(A1)*(DCOS(I1)-DCOS(I2))/KI
    N2= N1 + DCOS(A1)*(DCOS(I1)-DCOS(I2))/KI
    V2= V1 + (DSIN(I2)-DSIN(I1))/KI
    GO TO 5
C*****
C***** CASE 4: VARYING INCLINATION VARYING AZIMUTH
4   KI=(I2-I1)/(M2-M1)
    KA=KI*(A2-A1)/(DLOG(DTAN(I2/2.D0))-DLOG(DTAN(I1/2.D0)))
    IH=(I1+I2)/2.D0
    AH=A1+(KA/KI)*(DLOG(DTAN(I2/2.D0))-DLOG(DTAN(I1/2.D0)))
    E2= E1 + ( DSIN(I1)*DSIN(A1) + 4.D0*DSIN((IH))*DSIN((AH))
*         + DSIN(I2)*DSIN(A2) ) * (I2-I1)/(6.D0*KI)
    N2= N1 + ( DSIN(I1)*DCOS(A1) + 4.D0*DSIN((IH))*DCOS((AH))
*         + DSIN(I2)*DCOS(A2) ) * (I2-I1)/(6.D0*KI)
    V2= V1 + (DSIN(I2)-DSIN(I1))/KI
C*****
C FINALIZING LOOP AND CALCULATION OF DOG-LEG SEVERITY
5   I1=I2
    A1=A2
    M1=M2
    E1=E2
    N1=N2
    V1=V2
    E(L)=E2
    N(L)=N2
    V(L)=V2
    KD(L)=100.D0*DSQRT(KI*KI+KA*KA)/DR
    GO TO 20
C *****
9   CONTINUE
    RETURN
    END

```

6- INPUT/OUTPUT FILE FORMAT

The complete listing of an output file is given here to show the format of the output file. The highlighted part of it represents the input file.

MD	Incl.	Azim.	TVD	NS	EW	DLS
9238.	60.04	61.78	8809.7	608.3	1146.5	0.67
9325.	59.98	61.15	8853.2	644.5	1212.6	0.63
9356.	60.27	62.17	8868.6	657.2	1236.3	3.00
9456.	61.36	62.49	8917.4	697.6	1313.7	1.13
9550.	62.00	62.19	8962.0	736.1	1386.9	0.74
9643.	62.77	62.44	9005.1	774.3	1459.9	0.86
9739.	62.17	62.13	9049.5	814.0	1535.2	0.69
9773.	62.17	62.43	9065.3	828.0	1561.9	0.78
9867.	63.32	62.67	9108.4	866.4	1636.1	1.24
9928.	61.93	62.34	9136.4	891.5	1684.1	2.33
9989.	62.57	62.38	9164.8	916.5	1731.9	1.05
10089.	62.41	61.85	9211.0	958.3	1810.2	0.50
10153.	62.90	61.99	9240.4	985.0	1860.3	0.79
10180.	63.30	62.18	9252.6	996.2	1881.6	1.61
10255.	63.50	62.20	9286.2	1027.5	1940.9	0.27

7- CONCLUSIONS AND RECOMMENDATIONS

The calculations results obtained from the curvature method in double precision were compared against the results obtained from the Schlumberger minimum curvature method used in Finder.

Discrepancies in results of up to 5 feet in TVD, 24 **feet** in northing displacement and **48 feet** in easting displacement were found. Therefore we recommend the implementation of our curvature method in Finder instead of the current minimum curvature method.

8- REFERENCES

- 1- Surveying theory and Practice by Raymond E. Davis.
- 2- CALCULUS by Thomas and Finney, 9th edition. 1996.
- 3- Drilling Engineering, A complete Well Planning Approach, By Adams Charrier.1980.
- 4- Planning the Directional Well, A calculation Method. By W. McMillan, SPE-AIME 1981.
- 5- An Improved Method for Computing Directional Surveys. By G. Wilson, SPE-AIME 1968.
- 6- Use of Curvature Method to Determine True Vertical Reservoir Thickness. By R. Rivero SPE-AIME, 1971.
- 7- Computerized Directional Well Planning for Dual-Target Objective. By E.Maidla, A. Cordovil, J Pereira, J Flacao. SPE, June 1999
- 8- Well Deviation Survey Calculations, Tangential and Curvature Methods. By Ziad Amer,2003.